
Chapter 1

Calculating an Exceptional
Machine (Extended Version)
Graham Hutton and Joel Wright1

Abstract: In previous work we showed how to verify a compiler for a small
language with exceptions. In this article we show how tocalculate, as opposed to
verify, an abstract machine for this language. The key step is the use of Reynold’s
defunctionalization, an old program transformation technique that has recently
been rejuvenated by the work of Danvy et al.

1.1 INTRODUCTION

Exceptions are an important feature of modern programming languages, but their
compilation has traditionally been viewedas an advanced topic. In previous work
we showed how the basic method of compiling exceptions usingstack unwind-
ing can be explained and verified using elementary functional programming tech-
niques [HW04]. In particular, we developed a compiler for a small language with
exceptions, together with a proof of its correctness.

In the formal reasoning community, however, one prefersconstructions to ver-
ifications [Bac03]. That is, rather than first writing the compiler and then sep-
arately proving its correctness with respect to a semantics for the language, it
would be preferable to try and calculate the compiler [Mei92] directly from the
semantics, with the aim of giving a systematicdiscovery of the idea of compiling
exceptions using stack unwinding, as opposed to a post-hoc verification.

In this article we take a step towards this goal, by showing how to calculate an
abstract machine for evaluating expressions in our language with exceptions. The
key step in the calculation is the use ofdefunctionalization, a program transforma-
tion technique that eliminates the use of higher-order functions, first introduced
by Reynolds in his seminal work on definitional interpreters [Rey72].

Despite being simple and powerful, defunctionalization seems to be somewhat
neglected in recent years. For example, it features in few modern courses, text-
books, and research articles on program transformation, and does not seem to be
as widely known and used as it should be. Recently, however, defunctionaliza-
tion has been rejuvenated by the work of Danvy et al, who show how it can be
applied in a variety of different areas, including the systematic design of abstract
machines for functional languages [DN01, ABDM03b, ADM04].

In this article, we show how Danvy’s approach can be used to calculate an
abstract machine for our language with exceptions. Moreover, the calculation
is rabbit free, in the sense that there are no Eureka steps in which one needs to
metaphorically pull a rabbit out of a hat — all the required concepts arise naturally
from the calculation process itself. The approach is based upon the work of Danvy
et al, but the emphasis on calculation and the style of exposition are our own.

The language that we use comprises just integer values, an addition operator, a
single exceptional value called throw, and a catch operator for this value [HW04].
This language does not provide features that are necessary for actual program-
ming, but itdoes provide just what we need for expository purposes. In particular,
integers and addition constitute a minimal language inwhich to consider compu-
tation using a stack, and throw and catch constitute a minimal extension in which
such computations can involve exceptions.

Our development proceeds in two steps,starting with the exception-free part
of the language to introduce the basic techniques, to which support for exceptions
is then added in the second step. All the programs are written in Haskell [Pey03],
and all the calculations are presented using equational reasoning.

1.2 ABSTRACT MACHINES

An abstract machine can be defined as a term rewriting system for executing
programs in a particular language, and is given by a set of rewrite rules that make
explicit how each step of execution proceeds. Perhaps the best known example is
Landin’s SECD machine for the lambda calculus [Lan64], which comprises a set
of rewrite rules that operate on tuples with four components that give the machine
its name, called the stack, environment, control and dump.

For a simpler example, consider a language in which programs comprise a
sequence of push and add operations on a stack of integers. In Haskell, such
programs, operations and stacks can be represented by the following types:

type Prog = [Op]
data Op = PUSH Int | ADD

type Stack = [Int]

An abstract machine for this language is given by defining two rewrite rules on
pairs of programs and stacks from the setProg × Stack:

〈 PUSH n : ops , s 〉 −→ 〈 ops , n : s 〉
〈 ADD :ops , n :m : s 〉 −→ 〈 ops , n +m : s 〉

The first rule states that push places a new integer on top of the stack, while the
second states that add replaces the top two integers on the stack by their sum. This
machine can be implemented in Haskellby an execution function that repeatedly
applies the two rules until this is no longer possible:

exec :: (Prog,Stack) → (Prog,Stack)
exec (PUSH n :ops,s) = exec (ops,n : s)
exec (ADD :ops,n :m : s) = exec (ops,n +m : s)
exec (p,s) = (p,s)

For example,exec ([PUSH 1,PUSH 2,ADD], [ ]) gives the result([ ], [3]). In the
remainder of this article, we will use the term abstract machine for such a func-
tional implementation of an underlying set of rewrite rules.

At this point, some readers may be wondering about the relationship between
abstract and virtualmachines. The difference is thatan abstract machine operates
directly on programs themselves, whereas a virtual machine operates on compiled
versions of programs [ABDM03b]. However, the difference is typically just one
of context, because a virtual machine forthe source language of a compiler usually
takes the form of an abstract machine for the target language.

1.3 ARITHMETIC EXPRESSIONS

As in our previous work [HW04], let us begin our development by considering a
simple language of expressions comprising integers and addition, whose seman-
tics is given by a function that evaluates an expression to its integer value:

data Expr = Val Int | Add Expr Expr

eval :: Expr → Int
eval (Val n) = n
eval (Add x y) = eval x + eval y

We will now calculate an abstract machine for this language, by making a series
of three transformations to the semantics.

Step 1 - Add continuations

At present, the order in which addition evaluates its argument expressions is de-
termined by the language in which the semantics is written, in this case Haskell.
The first step in producing an abstract machine is to make the order of evaluation
explicit in the semantics itself. A standard technique for achieving this aim is to
rewrite the semantics incontinuation-passing style [Rey72].

A continuation is a function that will be applied to the result of an evaluation.
For example, in the equationeval (Add x y) = eval x +eval y from our semantics,
when the first recursive call,eval x, is being evaluated, the remainder of the right-
hand side,+ eval y, can be viewed as a continuation for this evaluation, in the
sense that it is the function that will be applied to the result.

More formally, in the context of our semanticseval :: Expr → Int, a continu-
ation is a function of typeInt → Int that will be applied to the result of typeInt
to give a new result of typeInt. (This type can be generalised toInt → a, but we
don’t need the extra generality for our purposes here.) We capture the notion of
such a continuation using the following type definition:

type Cont = Int → Int

Our aim now is to define a new semantics,eval′, that takes an expression and
returns an integer as previously, but also takes a continuation that will be applied
to the resulting integer. That is, we seek to define a function

eval′ :: Expr → Cont → Int

such that:

eval′ e c = c (eval e)

At this point in most texts, a recursive definition foreval′ would normally be
written and then either proved to satisfy the above equation, or this be justified
by appealing to the correctness of a general continuation-passing transformation.
However, we prefer tocalculate the definition foreval′ directly from the above
equation, by the use of structural induction onExpr.

Case:Val n

eval′ (Val n) c
= { specification ofeval′ }

c (eval (Val n))
= { definition ofeval }

c n

Case:Add x y

eval′ (Add x y) c
= { specification ofeval′ }

c (eval (Add x y))
= { definition ofeval }

c (eval x +eval y)
= { abstraction overeval x }

(λn → c (n +eval y)) (eval x)
= { induction hypothesis forx }

eval′ x (λn → c (n + eval y))
= { abstraction overeval y }

eval′ x (λn → (λm → c (n +m)) (eval y))
= { induction hypothesis fory }

eval′ x (λn → eval′ y (λm → c (n +m)))

In conclusion, we have calculated the following recursive definition:

eval′ :: Expr → Cont → Int
eval′ (Val n) c = c n
eval′ (Add x y) c = eval′ x (λn → eval′ y (λm → c (n +m)))

That is, for an integer value we simply apply the continuation, while for an addi-
tion we evaluate the first argument and call the resultn, then evaluate the second
argument and call the resultm, and finally apply the continuation to the sum ofn
andm. In this manner, order of evaluation is now explicit in the semantics.

Note that we have ensured that addition evaluates its arguments from left-to-
right by first abstracting overeval x in the above calculation, and then abstracting
overeval y. It would be perfectly valid to proceed in the other direction, which
would result in right-to-left evaluation. Note also that our original semantics can
be recovered from our new semantics, by substituting the identity continuation
λn → n into the equation from whicheval′ was constructed. That is, our original
semanticseval can now be redefined as follows:

eval :: Expr → Int
eval e = eval′ e (λn → n)

Step 2 - Defunctionalize

We have now taken a step towards an abstract machine by making evaluation
order explicit, but in so doing have also taken a step away from such a machine by
making the semantics into a higher-order function. The next step is to regain the
first-order nature of the original semantics by eliminating the use of continuations,
but retaining the explicit order of evaluation that they introduced.

A standard technique for eliminating the use of functions as arguments isde-
functionalization [Rey72]. This technique is based upon the observation that we
don’t usually need the entire function-space of possible argument functions, be-
cause only a few forms of such functions are actually used in practice. Hence,
we can represent the argument functions that we actually need using a datatype,
rather than using the actual functions themselves.

In our new semantics, there are only three forms of continuations that are actu-
ally used, namely one to invoke the semantics, and two in the case for evaluating
an addition. We begin by separating out these three forms, by giving them names
and abstracting over their free variables. That is, we define three combinators for
constructing the required forms of continuations:

c1 :: Cont
c1 = λn → n

c2 :: Expr → Cont → Cont
c2 y c = λn → eval′ y (c3 n c)
c3 :: Int → Cont → Cont
c3 n c = λm → c (n +m)

At present we have just used anonymous namesc1, c2 andc3 for the combina-
tors, but these will be replaced by more suggestive names later on. Using these
combinators, our semantics can now be rewritten as follows:

eval′ :: Expr → Cont → Int
eval′ (Val n) c = c n
eval′ (Add x y) c = eval′ x (c2 y c)
eval :: Expr → Int
eval e = eval′ e c1

The next stage in applying defunctionalization is to define a datatype whose
values represent the three combinators that we have isolated:

data CONT = C1 | C2 Expr CONT | C3 Int CONT

The constructors of this datatype have the same types as the corresponding com-
binators, except that the new typeCONT plays the role ofCont:

C1 :: CONT
C2 :: Expr → CONT → CONT
C3 :: Int → CONT → CONT

The fact that values of typeCONT represent continuations of typeCont can be
formalised by defining a function that maps from one to the other:

apply :: CONT → Cont
apply C1 = c1
apply (C2 y c) = c2 y (apply c)
apply (C3 n c) = c3 n (apply c)

The name of this function derives from the fact that when its type is expanded
to apply :: CONT → Int → Int, it can be viewed as applying a representation of a
continuation to an integer to give another integer.

Our aim now is to define a new semantics,eval′′, that behaves in the same way
as our previous semantics,eval′, except that it uses values of typeCONT rather
than continuations of typeCont. That is, we seek to define a function

eval′′ :: Expr → CONT → Int

such that:

eval′′ e c = eval′ e (apply c)

As previously, we calculate the definition for the functioneval′′ directly from this
equation by the use of structural induction onExpr.

Case:Val n

eval′′ (Val n) c
= { specification ofeval′′ }

eval′ (Val n) (apply c)
= { definition ofeval′ }

apply c n

Case:Add x y

eval′′ (Add x y) c
= { specification ofeval′′ }

eval′ (Add x y) (apply c)
= { definition ofeval′ }

eval′ x (c2 y (apply c))
= { definition ofapply }

eval′ x (apply (C2 y c))
= { induction hypothesis forx }

eval′′ x (C2 y c)

In conclusion, we have calculated the following recursive definition:

eval′′ :: Expr → CONT → Int
eval′′ (Val n) c = apply c n
eval′′ (Add x y) c = eval′′ x (C2 y c)

However, the definition forapply still refers to the previous semanticseval′, via
its use of the combinatorc2. We calculate a new definition forapply that refers to
our new semantics instead by the use of case analysis onCONT .

Case:C1

apply C1 n
= { definition ofapply }

c1 n
= { definition ofc1 }

n

Case:C2 y c

apply (C2 y c) n
= { definition ofapply }

c2 y (apply c) n
= { definition ofc2 }

eval′ y (c3 n (apply c))
= { definition ofapply }

eval′ y (apply (C3 n c))
= { specification ofeval′′ }

eval′′ y (C3 n c)

Case:C3 n c

apply (C3 n c) m
= { definition ofapply }

c3 n (apply c) m
= { definition ofc3 }

apply c (n +m)

In conclusion, we have calculated the following new definition:

apply :: CONT → Int → Int
apply C1 n = n
apply (C2 y c) n = eval′′ y (C3 n c)
apply (C3 n c) m = apply c (n +m)

We have now eliminated the use of functions as arguments, and hence made
the semantics first-order again. But what about the fact thateval′′ andapply are
curried functions, and hence return functions as results? As is common practice,
we do not view the use of functions as results as being higher-order, as it is not
essential and can easily be eliminated if required by uncurrying.

Finally, our original semantics can be recovered from our new semantics by
redefiningeval e = eval′′ e C1, as can be verified by a simple calculation:

eval e
= { previous definition ofeval }

eval′ e (λn → n)
= { definition ofc1 }

eval′ e c1
= { definition ofapply }

eval′ e (apply C1)
= { specification ofeval′′ }

eval′′ e C1

Step 3 - Refactor

At this point, after making two transformations to the original semantics, the
reader may be wondering what we have actually produced? In fact, we now have
an abstract machine for evaluating expressions, but this only becomes clear after
we refactor the definitions, in this simple case by just renaming the components.
In detail, we renameCONT asCont, C1 asSTOP, C2 asEVAL, C3 asADD, eval′′

aseval, apply asexec, andeval asrun to give the following machine:

data Cont = STOP | EVAL Expr Cont | ADD Int Cont

eval :: Expr → Cont → Int
eval (Val n) c = exec c n
eval (Add x y) c = eval x (EVAL y c)
exec :: Cont → Int → Int
exec STOP n = n
exec (EVAL y c) n = eval y (ADD n c)
exec (ADD n c) m = exec c (n +m)
run :: Expr → Int
run e = eval e STOP

We now explain the four parts of the abstract machine in turn:

• Cont is the type ofcontrol stacks for the machine, containing instructions that
determine the behaviour of the machine after evaluating the current expres-
sion. The meaning of the three forms of instructions,STOP, EVAL andADD
will be explained shortly. Note that the type of control stacks could itself be
refactored as an explicit list of instructions, as follows:

type Cont = [Inst ]
data Inst = ADD Int | EVAL Expr

However, we prefer the original definition above because it only requires the
definition of a single type rather than a pair of types.

• eval evaluates an expression in the context of a control stack. If the expression
is an integer value, it is already fully evaluated, and we simply execute the
control stack using this integer as an argument. If the expression is an addition,
we evaluate the first argument,x, placing the instructionEVAL y on top of
the current control stack to indicate that the second argument,y, should be
evaluated once that of the first argument is completed.

• exec executes a control stack in the context of an integer argument. If the stack
is empty, represented by the instructionSTOP, we simply return the integer
argument as the result of the execution. If the top of the stack is an instruction
EVAL y, we evaluate the expressiony, placing the instructionADD n on top of
the remaining stack to indicate that the current integer argument,n, should be
added together with the result of evaluatingy once this is completed. Finally,
if the top of the stack is an instructionADD m, evaluation of the two arguments
of an addition is now complete, and we execute the remaining control stack in
the context of the sum of the two resulting integers.

• run evaluates an expression to give an integer, by invokingeval with the given
expression and the empty control stack as arguments.

The fact that our machine uses twomutually recursive functions,eval andexec,
reflects the fact that it has two states, depending upon whether it is being driven
by the structure of the expression (eval) or the control stack (exec). To illustrate
the machine, here is how it evaluates(2+3)+4:

run (Add (Add (Val 2) (Val 3)) (Val 4))
= eval (Add (Add (Val 2) (Val 3)) (Val 4)) STOP
= eval (Add (Val 2) (Val 3)) (EVAL (Val 4) STOP)
= eval (Val 2) (EVAL (Val 3) (EVAL (Val 4) STOP))
= exec (EVAL (Val 3) (EVAL (Val 4) STOP)) 2
= eval (Val 3) (ADD 2 (EVAL (Val 4) STOP))
= exec (ADD 2 (EVAL (Val 4) STOP)) 3
= exec (EVAL (Val 4) STOP) 5
= eval (Val 4) (ADD 5 STOP)
= exec (ADD 5 STOP) 4
= exec STOP 9
= 9

Note how the functioneval proceeds downwards to the leftmost integer in the ex-
pression, maintaining a trail of the pending right-hand expressions on the control
stack. In turn, the functionexec then proceeds upwards through the trail, transfer-
ring control back toeval and performing additions as appropriate.

Readers familiar with Huet’szipper data structure for navigating around ex-
pressions [Hue97] may find it useful to note that our typeCont is a zipper data
structure forExpr, specialised to the purpose of evaluating expressions. Moreover,
this specialised zipper arose naturally here by a process of systematic calculation,
and did not require any prior knowledge of this structure.

1.4 ADDING EXCEPTIONS

Now let us extend our language of arithmetic expressions with simple primitives
for throwing and catching an exception:

data Expr = . . . | Throw | Catch Expr Expr

Informally, Throw abandons the current computation and throws an exception,
while Catch x y behaves as the expressionx unless it throws an exception, in which
case the catch behaves as thehandler expressiony. To formalise the meaning of
these new primitives, we first recall theMaybe type:

data Maybe a = Nothing | Just a

That is, a value of typeMaybe a is eitherNothing, which we think of as an excep-
tional value, or has the formJust x for somex of typea, which we think of as a
normal value [Spi90]. Using this type, our original semantics for expressions can

be rewritten to take account of exceptions as follows:

eval :: Expr → Maybe Int
eval (Val n) = Just n
eval (Add x y) = case eval x of

Nothing → Nothing
Just n → case eval y of

Nothing → Nothing
Just m → Just (n +m)

eval (Throw) = Nothing
eval (Catch x y) = case eval x of

Nothing → eval y
Just n → Just n

We will now calculate an abstract machine from this extended semantics, by fol-
lowing the same three-step process as previously. That is, we first add continua-
tions, then defunctionalize, and finally refactor the definitions.

Step 1 - Add continuations

Because our semantics now returns a result of typeMaybe Int, the type of contin-
uations that we use must be modified accordingly:

type Cont = Maybe Int → Maybe Int

Our aim now is to define a new semantics

eval′ :: Expr → Cont → Maybe Int

such that:
eval′ e c = c (eval e)

That is, the new semantics behaves in the same way aseval, except that it applies
a continuation to the result. As previously, we calculate the definition foreval′
from the above equation by structural induction onExpr.

Case:Val n

eval′ (Val n) c
= { specification ofeval′ }

c (eval (Val n))
= { definition ofeval }

c (Just n)

Case:Throw

eval′ Throw c
= { specification ofeval′ }

c (eval Throw)
= { definition ofeval }

c Nothing

Case:Add x y

eval′ (Add x y) c
= { specification ofeval′ }

c (eval (Add x y))
= { definition ofeval }

c (case eval x of
Nothing → Nothing
Just n → case eval y of

Nothing → Nothing
Just m → Just (n +m))

= { distribution overcase }
case eval x of

Nothing → c Nothing
Just n → case eval y of

Nothing → c Nothing
Just m → c (Just (n +m))

= { abstraction overeval y }
case eval x of

Nothing → c Nothing
Just n → (λy′ → case y′ of

Nothing → c Nothing
Just m → c (Just (n +m)) (eval y))

= { induction hypothesis fory }
case eval x of

Nothing → c Nothing
Just n → eval′ y (λy′ → case y′ of

Nothing → c Nothing
Just m → c (Just (n +m)))

= { abstraction overeval x }
(λx′ → case x′ of

Nothing → c Nothing
Just n → eval′ y (λy′ → case y′ of

Nothing → c Nothing
Just m → c (Just (n +m)))) (eval x)

= { induction hypothesis forx }
eval′ x (λx′ → case x′ of

Nothing → c Nothing
Just n → eval′ y (λy′ → case y′ of

Nothing → c Nothing
Just m → c (Just (n +m))))

Case:Catch x y

eval′ (Catch x y) c
= { specification ofeval′ }

c (eval (Catch x y))
= { definition ofeval }

c (case eval x of
Nothing → eval y
Just n → Just n)

= { distribution overcase }
case eval x of

Nothing → c (eval y)
Just n → c (Just n)

= { induction hypothesis fory }
case eval x of

Nothing → eval′ y c
Just n → c (Just n)

= { abstraction over evalx }
(λx′ → case x′ of

Nothing → eval′ y c
Just n → c (Just n)) (eval x)

= { induction hypothesis forx }
eval′ x (λx′ → case x′ of

Nothing → eval′ y c
Just n → c (Just n))

The two distribution overcase steps in the above calculation rely on the fact that
the continuationc is strict (c ⊥ = ⊥), which is indeed the case for all the forms
of continuation that we use, as we will see in the next section. In conclusion, we
have calculated the following recursive definition:

eval′ :: Expr → Cont → Maybe Int
eval′ (Val n) c = c (Just n)
eval′ (Throw) c = c Nothing
eval′ (Add x y) c = eval′ x (λx′ → case x′ of

Nothing → c Nothing
Just n → eval′ y (λy′ → case y′ of

Nothing → c Nothing
Just m → c (Just (n +m))))

eval′ (Catch x y) c = eval′ x (λx′ → case x′ of
Nothing → eval′ y c
Just n → c (Just n))

In turn, our original semantics can be recovered by invoking our new semantics
with the identity continuation. That is, we have

eval :: Expr → Maybe Int
eval e = eval′ e (λx → x)

Step 2 - Defunctionalize

Our new semantics uses four forms of continuations, namely one to invoke the
semantics, two in the case for addition, and one in the case for catch. We define
four combinators for constructing these continuations:

c1 :: Cont
c1 = λx → x

c2 :: Expr → Cont → Cont
c2 y c = λx′ → case x′ of

Nothing → c Nothing
Just n → eval′ y (c3 n c)

c3 :: Int → Cont → Cont
c3 n c = λy′ → case y′ of

Nothing → c Nothing
Just m → c (Just (n +m))

c4 :: Expr → Cont → Cont
c4 y c = λx′ → case x′ of

Nothing → eval′ y c
Just n → c (Just n)

Note that the resulting continuations are all strict, in the first case by being the
identity function, and in the other three by being defined using pattern matching.
Using these combinators, our semantics can now be rewritten as follows:

eval′ :: Expr → Cont → Maybe Int
eval′ (Val n) c = c (Just n)
eval′ (Throw) c = c Nothing
eval′ (Add x y) c = eval′ x (c2 y c)
eval′ (Catch x y) c = eval′ x (c4 y c)
eval :: Expr → Maybe Int
eval e = eval′ e c1

We now define a datatype to represent the four combinators, together with an
application function that formalises the representation:

data CONT = C1 | C2 Expr CONT | C3 Int Cont | C4 Expr CONT

apply :: CONT → Cont
apply C1 = c1
apply (C2 y c) = c2 y (apply c)
apply (C3 n c) = c3 n (apply c)
apply (C4 y c) = c4 y (apply c)

Our aim now is to define a new semantics

eval′′ :: Expr → CONT → Maybe Int

such that:

eval′′ e c = eval′ e (apply c)

That is, the new semantics behaves in the same way aseval′, except that it uses
representations of continuations rather than actual continuations. We calculate the
definition foreval′′ by structural induction onExpr.

Case:Val n

eval′′ (Val n) c
= { specification ofeval′′ }

eval′ (Val n) (apply c)
= { definition ofeval′ }

apply c (Just n)

Case:Throw

eval′′ Throw c
= { specification ofeval′′ }

eval′ Throw (apply c)
= { definition ofeval′ }

apply c Nothing

Case:Add x y

eval′′ (Add x y) c
= { specification ofeval′′ }

eval′ (Add x y) (apply c)
= { definition ofeval′ }

eval′ x (c2 y (apply c))
= { definition ofapply }

eval′ x (apply (C2 y c))
= { induction hypothesis forx }

eval′′ x (C2 y c)

Case:Catch x y

eval′′ (Catch x y) c
= { specification ofeval′′ }

eval′ (Catch x y) (apply c)
= { definition ofeval′ }

eval′ x (c4 y (apply c))
= { definition ofapply }

eval′ x (apply (C4 y c))
= { induction hypothesis forx }

eval′′ x (C4 y c)

In conclusion, we have:

eval′′ :: Expr → CONT → Maybe Int
eval′′ (Val n) c = apply c (Just n)
eval′′ (Throw) c = apply c Nothing
eval′′ (Add x y) c = eval′′ x (C2 y c)
eval′′ (Catch x y) c = eval′′ x (C4 y c)

In turn, we calculate a new definition forapply by case analysis.

Case:C1

apply C1 x
= { definition ofapply }

c1 x
= { definition ofc1 }

x

Case:C2 y c

apply (C2 y c) x′
= { definition ofapply }

c2 y (apply c) x′
= { definition ofc2 }

case x′ of
Nothing → apply c Nothing
Just n → eval′ y (c3 n (apply c))

= { definition ofapply }
case x′ of

Nothing → apply c Nothing
Just n → eval′ y (apply (C3 n c))

= { specification foreval′′ }
case x′ of

Nothing → apply c Nothing
Just n → eval′′ y (C3 n c)

Case:C3 n c

apply (C3 n c) y′
= { definition ofapply }

c3 n (apply c) y′
= { definition ofc3 }

case y′ of
Nothing → apply c Nothing
Just m → apply c (Just (n +m))

Case:C4 y c

apply (C4 y c) x′
= { definition ofapply }

c4 y (apply c) x′
= { definition ofc4 }

case x′ of
Nothing → eval′ y (apply c)
Just n → apply c (Just n)

= { specification foreval′′ }
case x′ of

Nothing → eval′′ y c
Just n → apply c (Just n)

In conclusion, we have:

apply :: CONT → Maybe Int → Maybe Int
apply C1 x = x
apply (C2 y c) Nothing = apply c Nothing
apply (C2 y c) (Just n) = eval′′ y (C3 n c)
apply (C3 n c) Nothing = apply c Nothing
apply (C3 n c) (Just m) = apply c (Just (n +m))
apply (C4 y c) Nothing = eval′′ y c
apply (C4 y c) (Just n) = apply c (Just n)

Our original semantics can be recovered by invoking our new semantics with the
representation of the identity continuation:

eval :: Expr → Maybe Int
eval = eval′′ e C1

Step 3 - Refactor

We now rename the components in the same way as previously, and rename the
new combinatorC4 asHAND. This time around, however, refactoring amounts
to more than just renaming. In particular, we split the application function

apply :: Cont → Maybe Int → Maybe Int

into two separate application functions

exec :: Cont → Int → Maybe Int
unwind :: Cont → Maybe Int

such that:
apply c (Just n) = exec c n
apply c Nothing = unwind c

That is, exec deals with normal arguments, andunwind with exceptional argu-
ments. We calculate the definition forexec by structural induction onCont.

Case:STOP

exec STOP n
= { specification ofexec }

apply STOP (Just n)
= { definition ofapply }

Just n

Case:EVAL y c

exec (EVAL y c) n
= { specification ofexec }

apply (EVAL y c) (Just n)
= { definition ofapply }

eval y (ADD n c)

Case:ADD n c

exec (ADD n c) m
= { specification ofexec }

apply (ADD n c) (Just m)
= { definition ofapply }

apply c (Just (n +m))
= { induction hypothesis}

exec c (n +m)

Case:HAND y c

exec (HAND y c) n
= { specification ofexec }

apply (HAND y c) (Just n)
= { definition ofapply }

apply c (Just n)
= { induction hypothesis}

exec c n

We also calculate the definition forunwind by structural induction onCont.

Case:STOP

unwind STOP
= { specification ofunwind }

apply STOP Nothing
= { definition ofapply }

Nothing

Case:EVAL y c

unwind (EVAL y c)
= { specification ofunwind }

apply (EVAL y c) Nothing
= { definition ofapply }

apply c Nothing
= { induction hypothesis}

unwind c

Case:ADD n c

unwind (ADD n c)
= { specification ofunwind }

apply (ADD n c) Nothing
= { definition ofapply }

apply c Nothing
= { induction hypothesis}

unwind c

Case:HAND y c

unwind (HAND y c)
= { specification ofunwind }

apply (HAND y c) Nothing
= { definition ofapply }

eval y c

In conclusion, we have constructed the following machine:

data Cont = STOP | EVAL Expr Cont |
ADD Int Cont | HAND Expr Cont

eval :: Expr → Cont → Maybe Int
eval (Val n) c = exec c n
eval (Throw) c = unwind c
eval (Add x y) c = eval x (EVAL y c)
eval (Catch x y) c = eval x (HAND y c)
exec :: Cont → Int → Maybe Int
exec STOP n = Just n
exec (EVAL y c) n = eval y (ADD n c)
exec (ADD n c) m = exec c (n +m)
exec (HAND c) n = exec c n

unwind :: Cont → Maybe Int
unwind STOP = Nothing
unwind (EVAL c) = unwind c
unwind (ADD c) = unwind c
unwind (HAND y c) = eval y c

run :: Expr → Maybe Int
run e = eval e STOP

We now explain the three main functions of the abstract machine:

• eval evaluates an expression in the context of a control stack. The cases for
integer values and addition are as previously. If the expression is a throw, we
unwind the stack seeking a handler expression. If the expression is a catch, we
evaluate its first argument,x, andmark the stack with the instructionHAND y
to indicate that its second argument, the handlery, should be used if evaluation
of its first produces an exceptional value.

• exec executes a control stack in the context of an integer argument. The first
three cases are as previously, except that if the stack is empty the resulting
integer is tagged as a normal result value. If the top of the stack is a handler
instruction, there is no need for the associated handler expression because a
normal integer result has already been produced, and weunmark the stack by
popping the handler and then continue executing.

• unwind executes the control stack in the context of an exception. If the stack is
empty, the exception is uncaught and we simply return the exceptional result
value. If the top of the stack is an evaluation or an addition instruction, there
is no need for their arguments because a handler is being sought, and we pop
them from the stack and then continue unwinding. If the top of the stack
is a handler instruction, we catch the exception by evaluating the associated
handler expression in the context of the remaining stack.

Note that the idea of marking, unmarking, and unwinding the stack arose di-
rectly from the calculations, and did not require any prior knowledge of these
concepts. It is also interesting to note that the above machine produced by cal-
culation is both simpler and more efficient that those we had previously designed
by hand. In particular, our previous machines did not make a clean separation be-
tween the three concepts of evaluating an expression (eval), executing the control
stack (exec) and unwinding the control stack (unwind).

To illustrate our machine, here is how it evaluates 1+(catch (2+ throw) 3):

run (Add (Val 1) (Catch (Add (Val 2) Throw) (Val 3)))
= eval (Add (Val 1) (Catch (Add (Val 2) Throw) (Val 3))) STOP
= eval (Val 1) (EVAL (Catch (Add (Val 2) Throw) (Val 3)) STOP)
= exec (EVAL (Catch (Add (Val 2) Throw) (Val 3)) STOP) 1
= eval (Catch (Add (Val 2) Throw) (Val 3)) (ADD 1 STOP)
= eval (Add (Val 2) Throw) (HAND (Val 3) (ADD 1 STOP))
= eval (Val 2) (EVAL Throw (HAND (Val 3) (ADD 1 STOP)))
= exec (EVAL Throw (HAND (Val 3) (ADD 1 STOP))) 2
= eval Throw (ADD 2 (HAND (Val 3) (ADD 1 STOP)))
= unwind (ADD 2 (HAND (Val 3) (ADD 1 STOP)))
= unwind (HAND (Val 3) (ADD 1 STOP))
= eval (Val 3) (ADD 1 STOP)
= exec (ADD 1 STOP) 3
= exec STOP 4
= 4

That is, the machine first proceeds normallyby transferring control back and for-
ward between the functionseval andexec, until the exception is encountered, at
which point the control stack is unwound to find the handler expression, and the
machine then proceeds normally once again.

1.5 FURTHER WORK

We have shown how an abstract machine for a small language with exceptions
can be calculated in a systematic way from a semantics for the language, using
a three-step process of adding continuations, defunctionalizing, and refactoring.
Moreover, the calculations themselves are straightforward, only requiring the ba-
sic concepts of structural induction and case analysis.

Possible directions for further work include exploring the impact of higher-
level algebraic methods (such as monads [Wad92] and folds [Hut99]) on the cal-
culations, mechanically checking the calculations using a theorem proving system
(for example, see [Nip04]), factorising the abstract machine into the composition
of a compiler and a virtual machine [ABDM03a], and generalising the underlying
language (we are particularly interested in the addition of interrupts.)

